


Queens



This classic puzzle only needed a good display that would do the job. On CITY ESCAPE 2 graphics were shown on 5 fields. I need 8 fields in a row here, but I don't need 2 graphics. With the display of CITY ESCAPE as base I could code the display of QUEENS. The game itself was easy to code, enough room left so no optimizations done or needed.

```
; 8 queens
; place 8 queens on a chessboard without attack

? * TORNADO *

                ORG  #4009                ;#4009
                DUMP  49161

qow             EQU  udgs*256/256
qob             EQU  qow+32
erq             EQU  qob+32
fb             EQU  fakeboard*256/256

basic           LD    B,3
L400B           JR    init0

                DEFB  236
                DEFB  212,28,126
                DEFB  143,0,18

eline           DEFW  last
chadd           DEFW  last-1
xptr            DEFW  0
stkbot          DEFW  last
stkend          DEFW  last
berg            DEFB  0
mem             DEFW  0                ; not needed without fp
```

```

init1      EX    AF,AF'
           JP    init

lastk      DEFB  255,255,255
margin     DEFB  55

nxtlin     DEFW  basic

init0      XOR    A
qcnt       DEFB  254

flagx      DEFB  0

strlen     JR     init1

taddr      DEFW  3213
seed       DEFW  0
frames     DEFW  65535
coords     DEFB  0,0
prcc       DEFB  188
sposn      DEFB  33,24
cdflag     DEFB  64


hr         LD     HL,dfile+#8000
           LD     BC,#401
           LD     A,#1E
           LD     I,A
           LD     A,#FA
           CALL   #2B5                ; lowres screen

           LD     BC,scrdata-2
           EXX
           LD     E,#10
           LD     HL,#4380            ; the board
           LD     A,H
           LD     I,A


           LD     B,6
hr0        DJNZ   hr0

           CALL   hrentry              ; set the board


           LD     B,13
hr1        DJNZ   hr1
           LD     A,(HL)


           LD     D,2
dloop      LD     B,E
           LD     A,qow-2
sh8        INC    A
           INC    A
q8         CALL   show8                ; show remaining queens
           PUSH   HL
           POP    HL
           DEC    B
           JP     NZ,sh8

           LD     HL,q8+2
           LD     A,#80
           XOR    (HL)
           LD     (HL),A

```

```

DEC    D
JR     NZ,dloop

CALL   #292
CALL   #220
LD     IX,hr
JP     #2A4

show8   LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
        LD     R,A
        DEFW   0
show8end RET    P                ; always true

cloop   ADD    A,E                ; calc next boardline
        LD     L,A                ; save it
        NOP                    ; timing
        RET    Z                ; next highbyte is exit

hrentry LD     B,E                ; 16 lines per boardline
        EXX
        INC    BC
        LD     A,(BC)            ; get UDG-pointer
        LD     H,udgs/256        ; make address
        LD     L,A
        DEC    BC

bloop   LD     A,(BC)            ; get fieldposition
        LD     D,H                ; set highbyte right
        LD     E,A
        LDI                    ; copy udg to board
        LDI                    ; size 16 pixels

        EXX
        DEC    B                ; decrease linecounter
        LD     A,L                ; get datapointer
        JP     lbuf+#8000        ; do display

lbuf    LD     R,A
        DEFW   0,0,0,0
        DEFW   0,0,0,0
        JP     Z,cloop          ; calc next line
        EXX
        JP     blow

blow    PUSH    HL                ; timing
        POP     HL
        INC     C                ; undo old line
        INC     C
        JR     bloop            ; do same udg again

```

```

; the data to display the queens
DEFB 0
endscr DEFB fb,erq*256/256,fb
DEFB qob*256/256,fb
DEFB qow*256/256,fb
DEFB erq*256/256,fb
DEFB qob*256/256,fb
DEFB qob*256/256,fb
DEFB qow*256/256,fb
DEFB qow*256/256
scrdata EQU $

field LD A,B
AND 7
LD B,A
LD A,C
AND 7
LD C,A
PUSH BC
INC B
INC C
LD H,#43
LD A,#80-2-16
ADD A,C
ADD A,C
ffield ADD A,16
DJNZ ffield
LD L,A
POP BC
RET

qfind LD BC,scrdata-2
fqueen LD A,(BC)
CP L
RET Z ; found

nextq DEC BC
DEC BC
LD A,C
CP endscr*256/256-1
RET C ; not on board
JR fqueen

dlrs DEFB 0,"D"-k,"L"-k,"R"-k,"F"-k,0

start LD HL,dlrs
redef LD A,(lastk)
INC A
JR NZ,redef ; wait for keyup
redef1 LD BC,(lastk)
LD A,C
INC A
JR Z,redef1 ; wait for keydown
PUSH HL
CALL #7BD ; translate key
POP HL
LD (HL),A ; save keycode
INC HL
LD A,(HL)
LD (dir),A ; show next keyread
OR A ; test on end
JR NZ,redef
JR restart

```

```

won      LD    HL,text-1
shift    INC    HL
          PUSH  HL
          LD    DE,message
          LD    BC,32
          LDIR
          LD    HL,frames
          LD    A,(HL)
          SUB   10
wf3      CP    (HL)
          JR    NZ,wf3
          POP   HL
          LD    A,(HL)
          CP    endtext*256/256
          JR    NZ,shift

restart   LD    HL,scrdata-2
          LD    DE,show8+2
          LD    B,8
          LD    A,B
          LD    (qcnt),A
          XOR   A
cls      LD    (HL),fb
          LD    (DE),A
          INC   DE
          LD    (DE),A
          INC   DE
          INC   DE
          INC   DE
          DEC   HL
          DEC   HL
          DJNZ  cls

          LD    C,B

gameloop LD    E,8                ; 8 rows
          LD    SP,fakeboard      ; just reset the stack each loop
          LD    HL,scrdata-2

resq     LD    A,(HL)            ; get position on row
          BIT   1,A              ; odd position on row?
          JR    Z,noxor
          XOR   16               ; transfer to bit 4
noxor    AND   16               ; get bit 4 only
          ADD   A,A              ; make it 0 / 32
          ADD   A,qow            ; add udg queen on white
          INC   HL
          LD    (HL),A           ; set udg back
          DEC   HL
          DEC   HL
          DEC   HL
          DEC   E
          JR    NZ,resq          ; do full board

          LD    A,(qcnt)         ; test here so
          OR    A               ; all queens are shown
          JR    Z,won           ; ok on the board

          LD    E,2
flloop   CALL  field
          PUSH  BC
          CALL  qfind
          INC   BC              ; when not found

```

```

LD    A,(BC)
XOR   96                ; fake position is flashed, no problems
LD    (BC),A
POP   BC

PUSH  HL
LD    A,(HL)
CPL
LD    (HL),A
INC   L
LD    (HL),A

LD    HL,frames
LD    A,(HL)
SUB   5
wfr   CP    (HL)
JR    NZ,wfr
POP   HL

DEC   E
JR    NZ,flloop        ; flash on and flash off
PUSH  BC
LD    BC,(lastk)
LD    A,C
INC   A
CALL  NZ,#7BD
POP   BC
LD    HL,dlrs
CP    (HL)              ; up
INC   HL
JR    NZ,testd         ; test down
DEC   B
testd CP    (HL)         ; down
INC   HL
JR    NZ,testl
INC   B
testl CP    (HL)         ; left
INC   HL
JR    NZ,testr
DEC   C
testr CP    (HL)         ; right
INC   HL
JR    NZ,testf
INC   C
testf SUB   (HL)         ; fire
JR    NZ,gameloop

setqueen LD    (erfnd+1),A    ; reset error
LD    D,3
yloop DEC   D
DEC   D
LD    E,3
xloop DEC   E
DEC   E

PUSH  BC
LD    A,D
OR    E
JR    Z,outboard        ; no dy and no dx
onboard CALL field
PUSH  BC
CALL  qfind
JR    NZ,step           ; no queen found
INC   BC

```

```

LD    A,erq                ; signal error queen
LD    (BC),A
LD    (erfnd+1),A
step  POP BC                ; get x' and y'
LD    A,B
ADD   A,D                  ; do dy
LD    B,A
LD    A,C
ADD   A,E                  ; do dx
LD    C,A
OR    B
BIT   3,A                  ; x or y out of board?
JR    Z,onboard
outboard POP BC            ; get original x and y
INC   E
JR    NZ,xloop             ; not all dx done
INC   D
JR    NZ,yloop             ; not all dy done

CALL  field
LD    A,L
SUB   #80
LD    DE,scrdata
pfield DEC DE
DEC   DE
SUB   16
JR    NC,pfield

LD    A,(DE)
CP    fb
JR    Z,erfnd              ; check error on empty line
CP    L
JR    Z,clearq             ; clear on same field

erfnd LD    A,0
OR    A
JR    Z,setq

showerror LD    HL,frames    ; 1 sec of current board
LD    A,(HL)
SUB   50
wf2     CP    (HL)
JR    NZ,wf2
gl      JP    gameloop

setq    LD    A,L
LD    (DE),A
LD    HL,show8end+2
ferfld  DEC   HL
DEC   HL
DEC   HL
LD    (HL),#40
DEC   HL
LD    A,(HL)
OR    A
JR    NZ,ferfld            ; find erasefield
LD    (HL),#40

LD    HL,qcnt
DEC   (HL)

JR    gl

```

```

clearq    BIT    1,A                ; odd position on row?
          JR     Z,noxor2
          XOR    16                ; transfer to bit 4
noxor2    AND    16                ; get bit 4 only
          JR     Z,clhl
          LD     A,255
clhl      LD     (HL),A
          INC    HL
          LD     (HL),A
          LD     A,fb
          LD     (DE),A
          LD     HL,qcnt
          INC    (HL)

          LD     HL,show8-1
fsetfld   INC    HL
          INC    HL
          INC    HL
          XOR    A
          LD     (HL),A
          INC    HL
          CP     (HL)
          JR     Z,fsetfld
          LD     (HL),A
          JR     gl

text      DEFB   60,42,49,49,0,41,52,51,42,27 ; "WELL DONE."
          DEFB   0                ; " "
          DEFB   62,52,58,0,43,52,58,51,41,0 ; "YOU FOUND "
          DEFB   38,0,56,52,49,58,57,46,52,51 ; "A SOLUTION"
          DEFB   27,0,57,55,62,0,43,46,51,41 ; ". TRY FIND"
          DEFB   46,51,44,0,38,51,52,57,45 ; "ING ANOTH"
          DEFB   42,55,27          ; "ER."
          DEFB   118,118
endtext   DEFB   118

k         EQU    27
dfile     DEFB   118
          DEFB   22,22,22,0
          DEFB   36,0,"Q"-k,"U"-k,"E"-k,"E"-k,"N"-k,"S"-k,0
          DEFB   "O"-k,"N"-k,0,"A"-k,0,"C"-k,"H"-k,"E"-k
          DEFB   "S"-k,"S"-k,"B"-k,"O"-k,"A"-k,"R"-k,"D"-k
          DEFB   0,22,22,22
          DEFB   118
message   DEFW   0,0,0,0,0,0,0
          DEFB   0
dir       DEFB   "U"-k
          DEFW   0,0,0,0,0,0,0,0
          DEFB   118,118
          DEFB   118,118

spa       EQU    #4380-98-$
          DEFS   spa                ; Stack area and coding room

fakeboard DEFW   0
; 2x udg
udgs      DEFB   0,0,0,0          ; queen on white
          DEFB   9,144,13,176
          DEFB   7,224,2,64
          DEFB   2,64,7,224

```



```

DEFB 4,32,4,32
DEFB 4,32,15,240
DEFB 24,24,31,248
DEFB 0,0,0,0

DEFB 255,255,224,7      ; queen on black
DEFB 233,151,237,183
DEFB 231,231,242,79
DEFB 242,79,247,239
DEFB 244,47,244,47
DEFB 228,39
DEFB 207,243,216,27
DEFB 223,251,192,3
DEFB 255,255

DEFB 170,170,64,5      ; error queen
DEFB 169,146,77,181
DEFB 167,226,82,69
DEFB 162,74,87,229
DEFB 164,42,84,37
DEFB 164,34,79,241
DEFB 152,26,95,249
DEFB 128,2,85,85

w      EQU 0
n      EQU 65535

; 2 lines of board, rest is copied
board  DEFW w,n,w,n,w,n,w,n
       DEFW n,w,n,w,n,w,n,w

init   LD    IX,hr
       LD    SP,fakeboard
       LD    H,#3F
       LD    D,#BF
       LD    E,L
       LDIR

       LD    HL,start
       PUSH HL
       LD    HL,board
       LD    DE,init
       LD    C,96
       JP    #19F9

vars   DEFB 128
last   EQU $

```